```
LaunchKernels[3]
```

CloudConnect: This version of Mathematica will no longer be supported for use with the Wolfram Cloud beginning Sat 1 Jul 2023. Please upgrade your license or contact us.

```
{KernelObject[1, local], KernelObject[2, local], KernelObject[3, local]}

(*Definition of the basic functions: quantum integers, factorials, 6j-symbol*)

qint[n_, r_] := N[Sin[2 Pi n / r] / Sin[2 Pi / r], 1760]

(*This notebook uses half integer colors rather than integer colors;
to convert to the notation of the paper all colors should be doubled*)
qfact[n_, r_] := If[n >= 0, Product[qint[i, r], {i, 1, n}], 0]

T[a_, b_, c_] := (a + b + c);
Q[a_, b_, c_, d_] := (a + b + c + d);

Delta[a_, b_, c_, r_] :=
 If[a + b + c <= r - 2 && a + b ≥ c && a + c ≥ b && b + c ≥ a && IntegerQ[a + b + c],
  Sqrt[qfact[(a + b - c), r] qfact[(a + c - b), r]
    qfact[(b + c - a), r] / qfact[T[a, b, c] + 1, r]], 0]

sixj[a_, b_, c_, d_, e_, f_, r_] :=
 I ^ (-2 (a + b + c + d + e + f)) Delta[a, b, c, r] Delta[a, e, f, r] Delta[b, d, f, r]
  Delta[c, d, e, r] Sum[(-1) ^ z qfact[z + 1, r] / (qfact[z - T[a, b, c], r]
       qfact[z - T[a, e, f], r] qfact[z - T[b, d, f], r] qfact[z - T[c, d, e], r]
       qfact[Q[a, b, d, e] - z, r] qfact[Q[a, c, d, f] - z, r] qfact[Q[b, c, e, f] - z, r]),
   {z, Max[{T[a, b, c], T[a, e, f], T[b, d, f], T[c, d, e]}],
    Min[{Q[a, b, d, e], Q[a, c, d, f], Q[b, c, e, f]}]}]


(*Definition of the Yokota invariant for a square pyramid*)

Pir4[a_, b_, c_, d_, e_, f_, g_, h_, r_] := ParallelSum[(-1) ^ (2 i) qint[2 i + 1, r]
   (sixj[a, i, c, f, b, e, r] sixj[a, i, c, g, d, h, r]) ^ 2, {i, 0, (r - 2) / 2, 1 / 2}]

(*Definition of the Yokota invariant for a pentagonal pyramid*)

Pir5[a_, b_, c_, d_, e_, f_, g_, h_, i_, j_, r_] :=
 ParallelSum[(-1) ^ (2 k) qint[2 k + 1, r] (-1) ^ (2 l) qint[2 l + 1, r]
   Abs[sixj[e, k, b, f, a, j, r] sixj[e, l, c, h, d, i, r] sixj[e, l, c, g, b, k, r]] ^ 2,
  {k, 0, (r - 2) / 2, 1 / 2}, {l, 0, (r - 2) / 2, 1 / 2}]

(*Yokota invariant with colors converging to the rectified square pyramid*)

Table6j4max[r_] := ParallelTable[sixj[Floor[r / 4], Floor[r / 4],
   Floor[r / 4], Floor[r / 4], Floor[r / 4], j, r], {j, 0, (r - 2) / 2, 1 / 2}]

Bulkpir4max[r_] := Module[{sixjarr},
  sixjarr = Table6j4max[r];
  Sum[(-1) ^ ((k - 1)) qint[k, r] Abs[sixjarr[[k]] ^ 2] ^ 2, {k, 1, (r - 2)}]]
```

```
(*Yokota invariant with colors converging to the regular ideal square pyramid*)

Table6j4id[r_] := ParallelTable[sixj[Floor[r / 8], Floor[r / 8], j,
   Floor[3 r / 16], Floor[3 r / 16], Floor[3 r / 16], r], {j, 0, (r - 2) / 2, 1 / 2}]

Bulkpir4id[r_] := Module[{sixjarr},
  sixjarr = Table6j4id[r];
  Sum[(-1) ^ ((k - 1)) qint[k, r] Abs[sixjarr[[k]] ^2] ^2, {k, 1, (r - 2)}]]

(*Yokota invariant with colors converging
 to the regular ideal pentagonal pyramid*)

Table6j5id1[r_] := ParallelTable[sixj[Floor[r / 10], Floor[r / 10],
   j, Floor[r / 5], Floor[r / 5], Floor[r / 5], r], {j, 0, (r - 2) / 2, 1 / 2}]

Table6j5id2[r_] :=
 ParallelTable[sixj[Floor[r / 10], j, k, Floor[r / 5], Floor[r / 5], Floor[r / 5], r],
  {j, 0, (r - 2) / 2, 1 / 2}, {k, 0, (r - 2) / 2, 1 / 2}]

Bulkpir5reg[r_] := Module[{sixjarr1, sixjarr2},
  sixjarr1 = Table6j5id1[r];
  sixjarr2 = Table6j5id2[r];
  Sum[(-1) ^ ((k - 1)) qint[k, r] (-1) ^ ((l - 1))
    qint[l, r] Abs[sixjarr1[[k]] sixjarr1[[l]] sixjarr2[[k, l]]] ^2,
   {k, 1, (r - 2)}, {l, 1, (r - 2)}]]

(*Yokota invariant with colors converging to the rectified pentagonal pyramid*)

Table6j5max[r_] :=
 ParallelTable[sixj[Floor[r / 4], Floor[r / 4], Floor[r / 4], Floor[r / 4], i, j, r],
  {i, 0, (r - 2) / 2, 1 / 2}, {j, 0, (r - 2) / 2, 1 / 2}]

Bulkpir5max[r_] := Module[{sixjarr},
  sixjarr = Table6j5max[r];
  Sum[(-1) ^ ((k - 1)) qint[k, r] (-1) ^ ((l - 1)) qint[l, r]
    Abs[sixjarr[[k, 2 Floor[r / 4] + 1]] sixjarr[[l, 2 Floor[r / 4] + 1]]
       sixjarr[[k, l]]] ^2, {k, 1, (r - 2)}, {l, 1, (r - 2)}]]

(*Example of calculation and timing for r=201;
all calculations done on a Dell XPS 13 with Intel(R) Core(TM) i5-
 7200U CPU@2.50GHz processors*)

AbsoluteTiming[Bulkpir4id[201] // N]
```

$\{4.2694, -1.07072 \times 10^{45}\}$

```
AbsoluteTiming[Bulkpir4max[201] // N]
```

$\{21.8521, -4.52791 \times 10^{157}\}$

```
AbsoluteTiming[Bulkpir5reg[201] // N]
```

$\{564.645, 2.16432 \times 10^{70}\}$

```
AbsoluteTiming[Bulkpir5max[201] // N]
```

$\{2254.54, 1.2807 \times 10^{213}\}$